
Selectarea elementului maxim
Fie urmãtoarele enunþuri de probleme!

P15. Într-un spital s-a mãsurat temperatura fiecãrui bol-
nav internat. Sã se stabileascã cea mai mare tempera-
turã mãsuratã!

P16. Având la dispoziþie un tabel cu numele elevilor parti-
cipanþi la un concurs, sã se stabileascã acel nume cu
care va începe tabelul nominal (în ordine alfabeticã)!

P17. Într-o clasã se face o statisticã, privind veniturile fa-
miliilor elevilor. Sã se determine venitul minim, exis-
tent printre venituri!

Observãm cã rezolvãrile acestor probleme - probabil -
vor fi asemãnãtoare, deoarece în fiecare se cere stabilirea
unei valori care este fie cea mai mare, fie cea mai micã în
ºirul dat. Bineînþeles, pot exista mai multe astfel de valori
egale cu valoarea minimã sau maximã. Dar în problemele
considerate nu se cere, de exemplu, sã numãrãm câþi bol-
navi au acea temperaturã maximã ºi nu ne intereseazã nici
numele acelui bolnav care este cel mai "fierbinte". Subal-
goritmul care rezolvã aceastã problemã (în termeni gene-
rali) va lucra cu o buclã de tip pentru, deoarece fiecare va-
loare trebuie prelucratã. În pseudocodul urmãtor lucrãm
cu entitãþile:

n: întreg {numãrul elementelor ºirului de prelucrat}
x: tablou(1..n): tip element {elementele ºirului dat}
val: tip element {rezultatul}

subalgoritm Maxim(n,x,val):

val ← x[1]

pentru i ← 2, n executã:

dacã val < x[i] atunci

val ← x[i]

sfârºit dacã

sfârºit pentru

sfârºit subalgoritm

Înainte de a trece la rezolvarea problemelor enunþate,
sã analizãm pseudocodul subalgoritmului. De ce nu am
iniþializat valoarea maximului (variabila val) cu 0, gândin-
du-ne cã sigur va exista un bolnav cu temperaturã poziti-
vã, deci nu existã pericolul ca maximul sã rãmânã 0? În re-
zolvarea problemei P15 am putea lucra cu aceastã valoare
iniþialã, dar dacã nu ar fi vorba de temperatura unor bol-
navi, ci de temperatura aerului în timpul unei luni într-o
iarnã geroasã, ar fi posibil sã nu avem nici o valoare  pozi-
tivã, ºi astfel maximul ar rãmâne 0, pãcãlindu-ne... Rezultã
cã se recomandã iniþializarea maximului (minimului) cu o
valoare existentã sigur în ºir, ºi, deoarece, de regulã, prelu-
crãm ºirurile începând cu primul element, cel mai firesc
este ca în iniþializarea respectivã sã folosim aceastã valoare. 
Sã analizãm oportunitatea de a implementa acest subal-
goritm cu o procedurã sau cu o funcþie. Avem de deter-
minat o singurã valoare, deci funcþia "ne face cu ochiul".
Dacã am folosi o funcþie denumitã "val", în instrucþiunea
if am avea un apel recursiv nedorit, deci ne-ar mai trebui
o variabilã suplimentarã, iar la sfârºit am atribui valoarea
acesteia identificatorului de funcþie. Presupunem cã în pro-
gramul apelant am declarat tipul:
type temp = array[1..50] of Real;

procedure P15(n:Byte; T:temp; var max:Real);

var i:Byte;

begin

max:=T[1];

for i:= 2 to n do

if max < T[i] then

max:=T[i]

end;

Având în vedere simplitatea implementãrii probleme-
lor P16 ºi P17, nu le vom detalia. În schimb, le vom modi-
fica enunþul, astfel încât sã vedem modul de rezolvare dacã
se cere ºi poziþia pe care se aflã minimul cerut. De exem-

Teorie ºi practicã!

TEOREME de programare
Clara Ionescu

Sfãtuim programatorii începãtori (dar ºi pe cei care se considerã "avansaþi") sã
respecte aceste modele de rezolvare a diferitelor probleme, considerate
clasice ºi care apar frecvent ca subprobleme în multe probleme de rezolvat.
Astfel, vor putea depana mult mai simplu programele scrise, le vor "citi" uºor ºi
peste mult timp, dar le vor înþelege ºi alþii. Nu este doar o recomandare din
partea GInfo, ci ºi a majoritãþii firmelor dezvoltatoare de software...

plu, în problema P16 se cere "cel mai mic nume". Sã ne
imaginãm cã în problemã nu se cere numele, ci doar indi-
cele în ºirul dat al aceluia care va ajunge în tabelul nominal
pe prima poziþie! Este evident cã rezultatul va fi nu o va-
loare datã în ºir (de data aceasta o valoare de tip string),
ci un numãr de ordine.  Presupunem cã în programul ape-
lant am declarat tipul:
type tabel = array[1..30] of string;

procedure P16_1(n:Byte; t:tabel; var ind:Byte);

var i:Byte;

min:string;

begin

min := t[1];   {presupunem cã min este primul}

ind := 1;     {dacã min rãmâne primul, ind e 1} 

for i := 2 to n do

if min > t[i] then begin

min := t[i];          {cel mai mic nume}

ind := i            {indicele acestuia}

end

end;

Nu ne-am pus problema cu privire la "care" minim ne
intereseazã, (respectiv, primul gãsit, al doilea sau ultimul)
deoarece într-un grup de elevi, probabil nu vor exista doi
copii cu nume identic. Dar dacã transpunem problema la
un ºir de cuvinte luate dintr-un text oarecare, atunci este
posibil sã existe mai multe asemenea valori minime. Ob-
servãm cã, dacã am utiliza acest subprogram în cazul res-
pectiv, acesta ne-ar furniza indicele primului element
având valoarea minimã. Dacã ni s-ar cere ultimul astfel de
element, atunci ar fi suficient sã schimbãm operatorul re-
laþional ">" în ">=". Dar se mai poate schimba ceva pen-
tru a scrie un subprogram care sã fie optimizat la "sânge".
Vom rezolva problema P16, modificând implementarea în
acest sens.
procedure P16_m(n:Byte; t:tabel; var min:Byte);

var i:Byte;

begin

min := 1;     {presupunem cã min este primul}

for i := 2 to n do

if t[min] >= t[i] then

min := i;

end;

Dacã, dupã apelarea subprogramului vrem sã ºtim ºi
numele elevului (având valoare minimã) vom cere afiºarea
elementului t[min].

Sã analizãm problema P17 într-o variantã modificatã,
în care ni s-ar cere numele tuturor copiilor care trãiesc în
familii având acelaºi venit minim. Rezolvarea acestei pro-
bleme se realizeazã în douã etape. Mai întâi stabilim mini-
mul în ºirul veniturilor, apoi parcurgem din nou întreg
ºirul ºi afiºãm elementele (numele copiilor) în cazul cãrora
venitul familiei este egal cu minimul gãsit în prima etapã.
În acest program trebuie sã declarãm tipurile utilizator:

type elev = record

nume : string;

venit : Real

end;

elevi = array[1..30] of elev;

procedure P17_1(n:Byte; e:elevi; var j:Byte;

var rez:elevi);

var i:Byte;

min:Real;

begin

min := e[1].venit;

for i := 2 to n do

if min > e[i].venit then

min := e[i].venit;

j := 0;

for i := 1 to n do

if min = e[i].venit then begin

Inc(j);

rez[j] := e[i];    {atribuire la nivel}

end                       {de articol}

end;

Dacã nu ne place faptul cã trebuie sã parcurgem de
douã ori ºirul, se poate scrie un subprogram care rezolvã
aceeaºi problemã cu o singurã parcurgere a lui: 
procedure P17_2(n:Byte; e:elevi; var j:Byte;

var rez:elevi);

var i:Byte;

min:Real;

begin

min := e[1].venit;

j := 1;

rez[j] := e[1];

for i := 2 to n do

if min > e[i].venit then begin

j := 1;

rez[j] := e[i]; 

min := e[i].venit

end else 

if min = e[i].venit then begin

Inc(j);

rez[j] := e[i]; 

end

end;

Dar astfel am ajuns la o problemã de selectare a unor
elemente dintr-un ºir dat pe baza unei proprietãþi. Aici
proprietatea a fost simplã: elementele selectate au avut un
câmp egal cu minimul din ºirul veniturilor. Pe baza obser-
vaþiei avem o nouã teoremã de programare.

Selectarea elementelor
Acest model de algoritm pe de o parte seamãnã cu cãuta-
rea liniarã, pe de altã parte cu numãrarea. Aici trebuie sã
furnizãm poziþiile a mai multor elemente, având o propri-
etate datã, deci nu este suficient sã le numãrãm.

subalgoritm Selectare_1(n,X,buc,Y):

buc ← 0

pentru i ← 1, n executã:

dacã P(X[i]) atunci

buc ← buc + 1

Y[buc] ← i

sfârºit dacã

sfârºit pentru

sfârºit subalgoritm

Se observã cã pentru pãstrarea rezultatului vom avea
nevoie de un tablou nou în cazul cãruia nu cunoaºtem
exact câte elemente va cuprinde. Dar, din moment ce nu
este exclus ca fiecare element al ºirului dat sã aibã propri-
etatea cerutã, acesta va fi declarat având atâtea elemente
câte are ºirul dat. Modelul de mai sus "adunã" indicii ele-
mentelor având proprietatea P în vectorul Y, ºi totodatã
obþine ºi numãrul acestora în buc. Dacã nu avem nevoie de
elementele respective colecþionte într-un vector nou, de-
oarece trebuie doar sã le furnizãm ca rezultat (de exemplu,
afiºându-le) algoritmul este ºi mai simplu:
subalgoritm Selectare_2(n,X):

pentru i ← 1, n executã:

dacã P(X[i]) atunci

scrie X[i] {sau i,în funcþie de cerinþe}

sfârºit dacã

sfârºit pentru

sfârºit subalgoritm

Vom da ºi a treia variantã a acestui algoritm. În aceasta,
considerãm cã dupã selectarea elementelor cãutate nu mai
avem nevoie de ºirul dat. Rezultã cã, în loc sã folosim un
al doilea ºir pentru elementele selectate, putem folosi chiar
ºirul dat. Pseudocodul acestui algoritm este urmãtorul:
subalgoritm Selectare_3(n,X,buc,Y):

buc ← 0

pentru i ← 1, n executã:

dacã P(X[i]) atunci

buc ← buc + 1

X[buc] ← X[i]

{sau i, în funcþie de cerinþe}

sfârºit dacã

sfârºit pentru

sfârºit subalgoritm

Oare cum rezolvãm problema, dacã se cere pãstrarea,
nu a elementelor având proprietatea P, ci a acelora care nu
au aceastã proprietate? Vom nega condiþia din if, veþi rãs-
punde! Aºa este. Dar dacã vrem sã pãstrãm aceste elemen-
te "nemiºcate" pe poziþiile lor? În acest caz, putem anula
prezenþa acestor elemente, suprascriindu-le cu o valoare
specialã, aleasã în acest scop. Astfel, vom avea un algoritm
care realizeazã o selectare pe loc. Aceastã soluþie va fi avan-
tajoasã doar dacã "evitarea" prelucrãrii acestor elemente,
având valoarea aleasã va fi mai simplu de realizat decât
verificarea proprietãþii P. 

subalgoritm Selectare_4(n,X):

pentru i ← 1, n executã:

dacã P(X[i]) atunci

X[i] ← valoare

sfârºit dacã

sfârºit pentru

sfârºit subalgoritm

Sã considerãm câteva enunþuri de probleme:
P18. Sã se realizeze un tabel cu fetele înscrise la olimpiada

de informaticã!
P19. Dat fiind un numãr natural, sã se genereze ºirul tutu-

ror divizorilor sãi!
P20. Fie ºirul mediilor semestriale ale elevilor unei clase în

care s-a comis o eroare, în sensul cã s-au calculat ºi
mediile copiilor care au rãmas corigenþi. Ar trebui
modificat ºirul, astfel încât media elevilor corigenþi sã
fie înlocuitã cu 0!

Rezolvarea problemei P18 se poate realiza aplicând al-
goritmul Selectare_1(n,X,buc,Y), presupunând cã ti-
pul sir desemneazã un tablou unidimensional de elemen-
te de tip record, unde printre câmpuri existã ºi câmpul
sex, care în cazul fetelor are valoarea 'F'.
procedure Tabel(n:Byte; elevi:sir; 

var nr:Byte; var fete:sir);

var i:Byte;

begin

nr := 0;

for i := 1 to n do

if elevi[i].sex = 'F' then begin

Inc(nr);

fete[nr] := elevi[i]

end

end;

Rezolvarea problemei P20 aparent este mai compli-
catã, dar numai din cauza cã proprietatea "corigent" nu
este memoratã explicit printre câmpurile articolelor, ci tre-
buie s-o stabilim pe baza mediilor pe discipline. Presupu-
nem cã programul conþine urmãtoarele declaraþii:
const nrdiscipline = 15;

type unelev = record

nume:string[30];

medii:array[1..15] of 3..10;

media:Real    {media generalã}

end;

type catalog = array[1..30] of unelev;

procedure Sterge(n:Byte; var c:catalog);

var i,j:Byte;

begin 

for i := 1 to n do begin 

j:=1;

while (j <= nrdiscipline) and

(c[i].medii[j] >= 5) do Inc(j);

if j <= nrdiscipline then

c[i].media := 0

end

end;


